

```
1  ' -----
2  '       From the book Inside Active Directory, ISBN 0-201-61621-1
3  '       Copyright (C) 2002 by Addison-Wesley
4  '       Script by Sakari Kouti (see http://www.kouti.com)
5  ' You have a royalty-free right to use, modify, reproduce and distribute
6  ' this script (and/or any modified version) in any way you find useful,
7  ' provided that you agree that Addison-Wesley or Sakari Kouti has no
8  ' warranty, obligations or liability for the script. If you modify
9  ' the script, you must retain this copyright notice.
10 ' -----
11 Option Explicit
12
13 Dim wshNetwork, wshShell, objFS, objServerService, objFileShare
14 Dim objDSE, objContainer, objVolume
15
16 Set wshNetwork = WScript.CreateObject("WScript.Network")
17 Set wshShell = Wscript.CreateObject("Wscript.Shell")
18
19 'Create the folder
20 Set objFS = CreateObject("Scripting.FileSystemObject")
21 Call objFS.CreateFolder("C:\ShareDemo")
22
23 'Share the folder
24 Set objServerService = GetObject("WinNT://" & _
25     wshNetwork.ComputerName & "/lanmanserver") _
26 Set objFileShare = objServerService.Create("fileshare", "ShareDemo")
27 objFileShare.Path = "C:\ShareDemo"
28 objFileShare.MaxUserCount = -1      'unlimited number of users
29 objFileShare.Description = "You should remove this demo share"
30 objFileShare.SetInfo
31
32 'Publish the share in Active Directory
33 Set objDSE = GetObject("LDAP://rootDSE")
34 Set objContainer = GetObject("LDAP://CN=Users," & _
35     objDSE.Get("defaultNamingContext"))
36 Set objVolume = objContainer.Create("volume", "CN=ShareDemo")
37 Call objVolume.Put("uNCName", _
38     "\\\" & wshNetwork.ComputerName & "\ShareDemo")
39 Call objVolume.Put("keywords", Array("demos", "sharing"))
40 objVolume.SetInfo
41
```